BGSCRIPT SCRIPTING LANGUAGE

DEVELOPER GUIDE

Thursday, 29 November 2012

Version 2.5



Copyright © 2001 - 2012 Bluegiga Technologies

Bluegiga Technologies reserves the right to alter the hardware, software, and/or specifications detailed herein at any time without notice, and does not make any commitment to update the information contained herein. Bluegiga Technologies assumes no responsibility for any errors which may appear in this manual. Bluegiga Technologies' products are not authorized for use as critical components in life support devices or systems.

Bluegiga Access Server, Access Point, APx4, AX4, BSM, iWRAP, BGScript and WRAP THOR are trademarks of Bluegiga Technologies.

The Bluetooth trademark and logo are registered trademarks and are owned by the Bluetooth SIG, Inc.

ARM and ARM9 are trademarks of ARM Ltd.

Linux is a trademark of Linus Torvalds.

All other trademarks listed herein belong to their respective owners.

TABLE OF CONTENTS

	Version history	
2.	What is BGScript?	. 6
	2.1 BGScript scripting language	. 6
3.	BGScript syntax	. 7
	3.1 Syntax	. 7
	3.1.1 Comments	
	3.1.2 Values	. 7
	3.1.3 Data types	
	3.1.4 Expressions	
	3.2 Commands	
	3.2.1 event <event_name> (<event_parameters>)</event_parameters></event_name>	
	3.2.2 if <expression> then [else] end if</expression>	
	3.2.3 while <expression> end while</expression>	. 10
	3.2.4 call <command name=""/> (<command parameters=""/>)[(response parameters)]	
	3.2.5 let <variable> = <expression></expression></variable>	
	3.2.6 sfloat(mantissa,exponent)	
	3.2.7 float(mantissa,exponent)	. 11
	3.2.8 memcpy(destination,source,length)	
	3.2.9 memcmp(buffer1,buffer2,length)	
4.	BGScript limitations	
	4.1 32-bit resolution	
	4.2 DIM variable size	
	4.3 Function support	
	4.4 Reading internal temperature meter disabled IO interrupts	
	4.5 Firmware can freeze if BGScript sends data to USB and no application reads it	. 13
	4.6 Performance	
5.	Example BGscripts	. 14
	5.1 Basics	. 14
	5.1.1 Catching system start-up	
	5.1.2 Catching Bluetooth connection event	
	5.1.3 Catching Bluetooth disconnection event	
	5.2 Hardware interfaces	
	5.2.1 ADC	
	5.2.2 I2C	
	5.2.3 IO	
	5.2.4 SPI	
	5.2.5 PWM	
	5.3 Timers	
	5.3.1 Continuous timer generated interrupt	
	5.3.2 Single timer generated interrupt	
	5.4 USB and UART endpoints	
	5.4.1 UART endpoint	
	5.4.2 USB endpoint	
	5.5 Attribute Protocol (ATT)	
	5.5.1 Catching attribute write event	
	5.6 Generic Attribute Profile (GATT)	
	5.6.1 Changing device name	
	5.6.2 Writing to local GATT database	
	5.7 PS store	
	5.7.1 Writing a PS keys	
	5.7.2 Reading a PS keys	
	5.8 Advanced scripting examples	
	5.8.1 Catching IO events and exposing them in GATT	
	5.9 DKBLE112 specific examples	
	5.9.1 Display initialization	
	5.9.2 FindMe demo	
	5.9.3 Temperature and battery readings to display	
	5.10 BGScript tricks	
	5.10.1 HEX to ASCII	
	5.10.2 UINT to ASCII	
6.	BGScript editors	. 41
	6.1 Notepad ++	. 41

| 6.1.1 Syntax high | nlight for BGSc | ript |
 |
 |
 |
 | |
 | 41 |
|------------------------|-----------------|------|------|------|------|------|--|------|------|------|------|------|------|------|------|----|
| 7. Contact information | | |
 |
 |
 |
 | |
 | 42 |

1 Version history

Version	Comments					
2.3	BGScript limitations updated with performance comments					
2.4	Added new features included in v.1.1 software. Small improvements made into BGScript examples Added a 4-channel PWM example					
2.5	Reading ADC does not disable IO interrupts					

2 What is BGScript?

2.1 BGScript scripting language

Bluegiga's *Bluetooth* Smart products allow application developers to create standalone devices without the need of a separate host. The *Bluetooth* Smart modules can run simple applications along the *Bluetooth* Smart stack and this provides a benefit when one needs to minimize the end product size, cost and current consumption. For developing these standalone *Bluetooth* Smart applications the Bluetooth Smart SDK provides a simple BASIC like BGScript scripting language. BGScript provides access to the same software and hardware interfaces as the BGAPI protocol. The BGScript code can be developed and compiled with free tools provided by Bluegiga.




Figure 1: Standalone application model

A BGScript code example:

```
# System Started
event system_boot(major, minor, patch, build, ll_version, protocol_version,
hw)

#Enable advertising mode
call gap_set_mode(gap_general_discoverable,gap_undirected_connectable)

#Enable bondable mode
call sm_set_bondable_mode(1)

#Start timer at 1 second interval (32768 = crystal frequency)
call hardware_set_soft_timer(32768)
end
```

A

When BGScript is used the BGAPI transport protocol is disabled.

3 BGScript syntax

3.1 Syntax

BGScript has BASIC-like syntax. Code is executed only in response to **event** messages. Lines are executed in order, starting from **event**-definition and finished at **return** or **end**. Each line is single command.

```
event hardware_io_port_status(delta, port, irq, state)
  tmp(0:1)=2
  tmp(1:1)=60*32768/delta

call attributes_write(xgatt_hr,2,tmp(0:2))
end
```

3.1.1 Comments

Anything after # to end of line is ignored.

```
X=1 #comment
```

3.1.2 Values

Values are decimal values. Hexadecimal values can be used by putting \$ before value

```
y=12
x=$ff
```

3.1.3 Data types

Datatypes can be defined globally using **dim** outside **event** block.

```
dim j
event hardware_soft_timer(handle )
  j=j+1
  call attributes_write(xgatt_counter,2,j)
end
```

Variables

Variables are signed 32-bit little-endian values. They need to be defined before use.

```
dim x
```

Usage

```
x=(2*2)+1
y=x+2
```

Constants

Constants are signed 32-bit little-endian values. They need to be defined before use.

```
const x=2
```

Usage

```
y=x+2
```

Buffers

Buffers hold 8-bit values. Buffers are not usable in mathematical expressions, but can be used to build some more complex data structures before inserting into attribute database.

Buffers need to be defined before use.

Maximum size of a buffer is 256 bytes.

```
dim u(10)
```

Usage

BUFFER(<expression>:<size>)

<expression> is index to first byte in buffer. Size is to specify how many bytes to use

```
u(0:1)=$a
u(1:2)=$123
#u(0:3) contains [$a][$23][$1]
```

3.1.4 Expressions

Expressions are given in infix notation.

```
x=(1+2)*(3+1)
```

Mathematical operators supported

Operation	Symbol
Addition:	+
Subtraction:	-
Multiplication:	*
Division:	1
Less than:	<
Less than or equal:	<=
Greater than:	>
Greater than or equal:	>=
Equals:	=
Not equals:	!=
	()

Bitwise operators supported:

Operation	Symbol
AND	&
OR	1
XOR	٨
Shift left	<<
Shift right	>>

Logical operators supported

Operation	Symbol
AND	&&
OR	

3.2 Commands

3.2.1 event <event_name> (<event_parameters>)

Code block defined between **event&end** will be run in response to a specific event. Execution will stop when reaching **end** or **return**.

```
event system_boot(buid,protocol_version,hw)
  call gap_set_mode(gap_general_discoverable,gap_undirected_connectable)
end
```

3.2.2 if <expression> then [else] end if

Condition can be tested with if. Commands between then and end if will be executed if <expression> is true.

```
if x<2 then
x=2
y=y+1
end if
```

If **else** is used, then if condition is success, commands between **then** and **else** will be executed. And if condition fails, commands between **else** and **end if** will be executed.

```
if x<2 then
  x=2
  y=y+1
else
  y=y-1
end if</pre>
```

3.2.3 while <expression> end while

Loops can be made using **while**. Command lines between **while** and **end while** will be executed while <expression> is true.

```
a=0
while a<10
a=a+1
end while
```

3.2.4 call <command name>(<command parameters>..)[(response parameters)]

Send API command and receive response. Command parameters can be given as expressions. Response parameters are variable names where response values will be loaded. Response parentheses and parameters can be omitted.

```
#xgatt_hr-attribute will receive 2 bytes from tmp buffer starting at index 0
and continuing 2bytes. Variable r will receive result response
call attributes_write(xgatt_hr,2,tmp(0:2))(r)
```

3.2.5 let <variable> = <expression>

Optional command to assign expression to variable

```
let a=1
let b=a+2
```

3.2.6 sfloat(mantissa, exponent)

Changes given mantissa and exponent in to 16bit IEEE 11073 SFLOAT value which has base-10. Conversion is done using following algorithm:

	Exponent	Mantissa
Length	4 bit	12 bit
Туре	2-complement	2-complement

Mathematically, the number generated by sfloat() is calculated as **<mantissa>** * **10^<exponent>**. The return value is a 2-byte uint8s array in the SFLOAT format. Here are some example parameters, and their resulting decimal sfloat values:

Mantissa	Exponent	Result (actual)
-105	-1	-10.5
100	0	100
320	3	320,000

Use the sfloat() function as follows, assuming that buf is already defined as a 2-byte uint8s array (or bigger):

buf will now contain the SFLOAT representation of -10.5.

Some reserved special purpose values:

- NaN (not a number)
 - exponent 0
 - mantissa 0x007FF
- **NRes** (not at this resolution)
 - exponent 0
 - mantissa **0x00800**
- Positive infinity
 - exponent 0
 - mantissa 0x007FE
- Negative infinity
 - exponent 0
 - mantissa 0x00802
- · Reserved for future use
 - exponent 0
 - mantissa 0x00801

3.2.7 float(mantissa, exponent)

Changes given mantissa and exponent in to 32bit IEEE 11073 SFLOAT value which has base-10. Conversion is done using following algorithm:

	Exponent	Mantissa
Length	8 bit	24 bit
Туре	signed integer	signed integer

Some reserved special purpose values:

- NaN (not a number)
 - exponent 0
 - mantissa 0x007FFFFF
- NRes (not at this resolution)

- exponent 0
- mantissa 0x00800000
- Positive infinity
 - exponent 0
 - mantissa 0x007FFFE
- Negative infinity
 - exponent 0
 - mantissa 0x00800002
- Reserved for future use
 - exponent 0
 - mantissa 0x00800001

3.2.8 memcpy(destination,source,length)

This is a supporting function for buffer handling. This function copies bytes from source buffer to destination buffer. Destination and source should not overlap.

```
dim dst(3)
dim src(4)
memcpy(dst(0),src(1),3)
```

3.2.9 memcmp(buffer1,buffer2,length)

This is a supporting function for buffer handling. This function compares *buffer1* with *buffer2*, length amount of bytes. The function returns 1 if the data matches.

```
\begin{array}{l} \text{dim } x(3) \\ \text{dim } y(4) \\ \text{if } \text{memcmp}(x(0),y(1),3) \text{ then} \\ \text{\#do something} \\ \text{end if} \end{array}
```

4 BGScript limitations

4.1 32-bit resolution

All operations in BGScript must be done with values that fit into 32 bits. The limitation affects for example long timer intervals. Since the soft timer has a 32.768kHz tick speed, it is possible in theory to have maximum interval of (2^32-1)/32768kHz = 36.4h, so if you need something to happen with longer periods you need to increment a dedicated counter in your script.



In particular with Bluetooth LE products, timer is 22 bits, so the maximum value with BLE112 is 2^22 = 4194304/32768Hz = 256 seconds, while with BLED112 USB dongle the maximum value is 2^22 = 4194304/32000Hz = 261 seconds

4.2 DIM variable size

The largest size of a DIM variable can be 256 bytes.

4.3 Function support

Currently BGScript doesn't support the use of functions.

4.4 Reading internal temperature meter disabled IO interrupts

Reading BLE112 internal temperature sensor value

call hardware_adc_read(14,3,0)

4.5 Firmware can freeze if BGScript sends data to USB and no application reads it

If the USB interface enabled and there is an USB cable connected to the BLE112 evaluation kit or BLED112 USB dongle, there needs to be an application reading the data. Otherwise the BGAPI messages will fill the buffers of the BLE112 and cause the firmware the freeze eventually.

4.6 Performance

BGScript has limited performance, which might prevent some applications to be implemented using BGscript. Typically executing a single line of BGScript will take 1-3ms.

5 Example BGscripts

This section contains useful BGScript examples.

5.1 Basics

This section contains very basic BGScript examples.

5.1.1 Catching system start-up

This example shows how to catch a system start-up. This event is the entry point to all BGScript code execution and can be compared to main() function in C.

System start-up

```
# Boot event listener
event system_boot(major,minor,patch,build,ll_version,protocol,hw)

# System started, enable advertising and allow connections
call gap_set_mode(gap_general_discoverable,gap_undirected_connectable)
...
end
```

5.1.2 Catching Bluetooth connection event

When a Bluetooth connection is received a connection_status(...) event is generated.

The example below shows how to enable advertisements to make the device connectable and how to catch a *Bluetooth* connection event.

Entering advertisement mode after disconnect

```
dim connected
# System start/boot listener
event system_boot(major,minor,patch,build,ll_version,protocol,hw)
   # Device is not connected yet
   connected = 0
   # Set advertisement interval to 20 to 30ms. Use all advertisement channels
   call gap_set_adv_parameters(32,48,7)
   # Start advertisement (generic discoverable, undirected connectable)
   call gap_set_mode(2,2)
end
# Connection event listener
event connection_status(connection, flags, address, address_type,
conn_interval, timeout, latency, bonding)
   # Device is connected.
   connected = 1
end
```

5.1.3 Catching Bluetooth disconnection event

When a Bluetooth connection is lost a connection_disconnected event is created.

Entering advertisement mode after disconnect

```
# Disconnection event
event connection_disconnected(handle, result)
   #connection disconnected, continue advertising
   call gap_set_mode(gap_general_discoverable,gap_undirected_connectable)
end
```

5.2 Hardware interfaces

This section contains basic examples to use hardware interfaces like I2C, SPI, AIO etc. from the BGScript.

5.2.1 ADC

ADC events can be cached with **hardware_adc_result(...)** event listener and the read operations on the other hand are called with **call hardware_adc_read(...)** function.

The example below shows how to read the internal temperature monitor and how to convert the value into Celsius

ADC read

```
dim celsius
dim offset
dim tmp(5)
# System boot event generated when the device is stared
event system_boot(major ,minor ,patch ,build ,ll_version ,protocol_version ,hw
   # Call ADC read.
   # 14 = internal temperature sensor
   # 3 = 12 effective bits
   # 0 = Internal 1.15V reference
   call hardware_adc_read(14,3,0)
end
# ADC event listener
event hardware_adc_result(input,value)
   # ADC value is 12 MSB
   celsius = value / 16
   # Calculate temperature
   # ADC*V_ref/ADC_max * T_coeff + offset
   celsius = (10*celsius*1150/2047) * 10/45 + offset
   # set flags according to Health Thermometer specification
   # 0 = Temperature in Celsius
   tmp(0:1)=0
   # Convert to float
   tmp(1:4)=float(celsius, -1)
end
```

5.2.2 I2C

BLE112 has a software I2C which uses fixed pins. For communicating over the I2C bus following hardware setup is needed:

- P1_6: I2C data
- P1_7: I2C clock



No UART or SPI can be used in channel 1 with alternative 2 configuration when I2C is used.

I2C operations

```
# Reading 2 bytes from device which has I2C address of 128.
# Result 0 indicates successful read.
call hardware_i2c_read(128,2)(result,data_len,data)
# Write to address 128 one byte (0xf5).
\ensuremath{\sharp} written indicates how many bytes were successfully written.
call hardware_i2c_write(128,1,"\xf5")(written)
```

5.2.3 IO

IO wake-up

When the device has no active tasks or timers running it can go to power mode 3 (PM3), which is the lower power mode consuming about 400nA. PM3 power save mode however requires an external wake-up using an IO pin.

The example here shows and IO interrupt can be used to wake up the device and start advertisements for 5 seconds and then go back to PM3.

Enabling and catching IO interrupts

```
# Boot event listener
event system_boot(major, minor, patch, build, ll_version, protocol_version,
   # Enable IO interrupts from PORT 0 PINs PO_0 and PO_1 on rising edge
   call hardware_io_port_config_irq(0,$3,0)
end
# HW interrupt listener
event hardware_io_port_status(delta, port, irq, state)
   # Configure advertisement parameters
   call gap_set_adv_parameters(40, 40, 7)
   # Start advertisements
  call gap_set_mode(2, 2)
   # Start a 5 second, one stop timer
   call hardware_set_soft_timer($27FFB, 0 ,1)
end
# Timer event listener
event hardware_soft_timer(handle)
   #Stop advertisements and allow the device to go to PM3
   call gap_set_mode(0, 0)
end
```

To enable PM3 and configure the wake-up pin the following configurations need to be used in the hardware.xml file.

```
<hardware>
  <sleeposc enable="true" ppm="30" />
  <wakeup pin enable="true" port="0" pin="0" />
  <usb enable="false" endpoint="none" /> <txpower power="15" bias="5" />
  <port index="0" tristatemask="0" pull="down" />
  <script enable="true" />
  <slow_clock enable="true" />
</hardware>
```

Writing IO status

The example below shows how to write the P0_0 status.

Enabling and catching IO interrupts

```
# Boot event listener
event system_boot(major ,minor ,patch ,build ,ll_version ,protocol_version ,hw
)

# Configure the PO_0 as output
call hardware_io_port_config_direction(0, 1)
# Enable PO_0 pin
call hardware_io_port_write(0, 1, 1)

# Start a 5 second, one stop timer
call hardware_set_soft_timer($27FFB, 0 ,1)
end

# Timer event listener
event hardware_soft_timer(handle)
# When timer expires disable PO_0 pin
call hardware_io_port_write(0, 1, 0)
end
```

5.2.4 SPI

Writing SPI

SPI interface can be used as a peripheral interface for example to connect to sensors like accelorometers or simple displays. The example below shows how to write data to SPI interface.

Writing to SPI

```
# Boot event listener
event system_boot(major, minor, patch, build, ll_version, protocol_version,
   # Writing 5 bytes to SPI
   call hardware_spi_transfer(0,5,"\x01\x02\x03\x04\x05")
   \mbox{\tt\#} Writeing a "Hello world\!" string to SPI
   call hardware_spi_transfer(0,12,"Hello world\!")
end
```

The following configurations need to be in the hardware.xml to enable the SPI interface and BGScript execution.

<hardware>

```
<usart channel="0" mode="spi_master" alternate="2" polarity="positive" phase="1"</pre>
endianness="msb" baud="57600" endpoint="none" />
  <script enable="true" />
</hardware>
```

Reading SPI

The example below shows how to read data from SPI interface. SPI interface returns you as many bytes as you write to it. In this example two (2) bytes are written to SPI interface and the return values return the read result. The read data is stored in the tmp-array and it has length on two (2) bytes.

Reading SPI interface

```
dim tmp(10)
dim result
dim channel
dim tlen
call hardware_spi_transfer(0,2,"\x01\x02")(result,channel,tlen,tmp(0))
```

The following configurations need to be in the hardware.xml to enable the SPI interface and BGScript execution.

<hardware>

```
<usart channel="0" mode="spi_master" alternate="2" polarity="positive" phase="1"</pre>
endianness="msb" baud="57600" endpoint="none" />
  <script enable="true" />
</hardware>
```

5.2.5 PWM

Generating PWM signals

In order to generate PWM signals output compare mode needs to be used. PWM output signals can be generated using the timer modulo mode and when channels 1 and 2 are in output compare mode 6 or 7.

For detailed instructions about PWM please refer to chapter 9.8 Output Compare Mode in CC2540 user guide.


In order to generate a 4 channel PWM signal the following example can be used.

A 4 channel PWM signal

```
# Boot event listener
event system_boot(major, minor, patch, build, ll_version, protocol_version,
call hardware timer comparator(1, 0, 6, 32000)
call hardware_timer_comparator(1, 1, 6, 16000)
call hardware_timer_comparator(1, 2, 6, 10000)
call hardware_timer_comparator(1, 3, 6, 8000)
call hardware_timer_comparator(1, 4, 6, 4000)
end
```

The example uses Timer 1 in alternate 2 configuration with four (4) PWM channels in pins p1.1, p1.0, p0.7 and p0.6

The following configurations need to be in the hardware.xml to enable the timer and BGScript execution.



Notice that PWMs do not work when the device is in a sleep mode.

5.3 Timers

This section describes how to use timers with BGscript.

5.3.1 Continuous timer generated interrupt

This example shows how to generate continuous timer generated interrupts

Enabling timer generated interrupts

```
# Boot event listener
event system_boot(major ,minor ,patch ,build ,ll_version ,protocol_version ,hw
)
    ...
    #Set timer to generate event every ls
    call hardware_set_soft_timer(32768, 1, 0)
    ...
end

#Timer event listener
event hardware_soft_timer(handle)
    #Code that you want to execute once per ls
    ...
end
```

Even with a soft timer running the module can enter sleep mode 2, in which power consumption is about 1µA. Sleep mode 3 is entered only if there are no timers running and the module is not having any scheduled radio activity.



One active timer

There can only be one timer running at the same time. Please stop the currently running timer by issuing **call hardware_set_soft_timer(0, {handle}, {singleshot})**) before launching the next one.

5.3.2 Single timer generated interrupt

The 2nd example shows how to set a timer, which is called only once. This is useful, when some action needs to be implemented only once, like the change of advertisement interval in Proximity profile.

In this example in the beginning the device advertises quickly, but after 30 seconds the advertisement interval is reduced, in order to save battery.

Using timer once

```
# Boot event listener
event system_boot(major,minor,patch,build,ll_version,protocol,hw)
   # Set advertisement parameters according to the Proximity profile
   # Min interval 20ms, max interval 30ms, use all 3 channels
   call gap_set_adv_parameters(32, 48, 7)
   # Enabled advertisement
   # Limited discovery, Undirected connectable
  call gap_set_mode(1, 2)
   # Start timer
   # single shot, 30 secods, timer handle = 1
   call hardware set soft timer($F0000, 1, 1)
end
# Timer event listener
event hardware_soft_timer(handle)
   # run the code only if timer handle is 1
   if handle = 1 then
     # Stop advertisement
      call gap_set_mode(0, 0)
      #Reconfigure parameters
      # Min interval 1000ms, max interval 2500ms, use all 3 channels
      call gap_set_adv_parameters(1600, 4000, 7)
      # Enabled advertisement
      # Limited discovery, Undirected connectable
      call gap_set_mode(1, 2)
   end if
end
```

5.4 USB and UART endpoints

This section describes the usage of endpoints, which can be used to send or receive data from interfaces like UART or USB.

5.4.1 UART endpoint

The example shows how to send data to USART1 endpoint from BGScript.

Writing to USB endpoint

```
# System start/boot listener
event system_boot(major,minor,patch,build,ll_version,protocol,hw)
  # Start continuous timer with 1 second interval. Handle ID 1
 # 1 second = $8000 (32.768kHz crystal)
 call hardware_set_soft_timer($8000, 1, 0)
end
# Timer event(s) listener
event hardware_soft_timer(handle)
  # 1 second timer expired
  if handle = 1 then
    call system endpoint_tx(5, 14, "TIMER EXPIRED\n")
 end if
end
```

The following configurations need to be in the hardware.xml to enable the UART interface and allow BGscript to access it.

```
<?xml version="1.0" encoding="UTF-8" ?>
 <hardware>
 <usart channel="1" alternate="1" baud="115200" endpoint="none" />
 <script enable="true" />
</hardware>
```

5.4.2 USB endpoint

The example shows how to send data to USB endpoint from BGScript.

Writing to USB endpoint

```
# System start/boot listener
event system_boot(major,minor,patch,build,ll_version,protocol,hw)
 # Start continuous timer with 1 second interval. Handle ID 1
 # 1 second = $8000 (32.768kHz crystal)
 call hardware_set_soft_timer($8000, 1, 0)
# Timer event(s) listener
event hardware_soft_timer(handle)
  # 1 second timer expired
 if handle = 1 then
   call system_endpoint_tx(3, 14, "TIMER EXPIRED\n")
 end if
end
```

The following configurations need to be in the hardware.xml to enable the USB interface and allow BGscript to access it.

```
<?xml version="1.0" encoding="UTF-8" ?>
 <hardware>
 <usb enable="true" endpoint="none" />
 <script enable="true" />
</hardware>
```

5.5 Attribute Protocol (ATT)

This section contains BGscript examples related to Attribute Protocol (ATT) events.

5.5.1 Catching attribute write event

The example shows to to catch an event when remote devices writes an attribute over a Bluetooth connection. A simple FindMe example is used where the remote device writes a single value to the local GATT database indicating the alert level.

Catching an attribute write

```
# Listen for GATT write events
event attributes_value(connection, reason, handle, value_len, value)

# Read the value and enable corresponding alert
level=value(0:1)
if level=0 then
    # TODO: Execute an action corresponding "No alert" status.
end if
if level=1 then
    # TODO: Execute an action corresponding "Mild alert" status.
end if
if level=2 then
    # TODO: Execute an action corresponding "High alert" status.
end if
end
```

5.6 Generic Attribute Profile (GATT)

This section shows examples how to manager the local GATT database.

5.6.1 Changing device name

The example below shows how to change the device name using BGScript.

In this example we use the following GATT database:

To write a new value into the characteristic defined in the **gatt.xm**l following code needs to be used. Please note that the **id** must be the same as in the **gatt.xm**l.

```
# Generate Friendly name in ASCII

name(0:1)=$42

name(1:1)=$47

name(2:1)=$53

name(3:1)=$63

name(4:1)=$72

name(5:1)=$69

name(6:1)=$70

name(7:1)=$74

#Write name to local GATT

call attributes_write(xgatt_name, 0, 7, name(0:7))
```

5.6.2 Writing to local GATT database

To write to the local GATT database you first need to define a characteristic under a service in your GATT database (**gatt.xml**). You also need to assign an **id** parameter for the characteristic, which can then be used in BGScript to write the value.

In this example we use the following GATT database:

To write a new value into the characteristic defined in the **gatt.xm**l following code needs to be used. Please note that the **id** must be the same as in the **gatt.xm**l.

script.bgs

```
#write 5 bytes from tmp array to attribute with offset 0
call attributes_write(xgatt_temperature_celsius,0,5,tmp(0:5))
```

5.7 PS store

These examples show how to read and write PS-keys.

5.7.1 Writing a PS keys

The example shows how to write an attribute written by a remote Bluetooth device into PS store.

Writing to PS store

```
# Check if remote device writes a value to the GATT and write it to a PS key
0x8000
# Catch an attribute write
event attributes_value(connection, reason, handle, offset, value_len,
value_data)

# Check if handle value 1 is written
if handle = 1
# Write attribute value to PS-store
call flash_ps_save($8000, value_len, value_data(0:value_len))
end if
end
```

1

PS keys from 8000 to 807F can be used for persistent storage of user data. Each key can store up to 32 Bytes.

5.7.2 Reading a PS keys

The example shows how to read a value from the local PS store and write it to GATT database.

Reading PS store

```
#Initialize a GATT value from a PS key, which is 2 bytes long
call flash_ps_load($8000)(result, len1, data1(0:2))
# Write the PS value to handle with ID "xgatt_PS_value"
call attributes_write(xgatt_PS_value, 0, len1, data1(0:len1))
```



PS keys from 8000 to 807F can be used for persistent storage of user data. Each key can store up to 32 Bytes.

5.8 Advanced scripting examples

This section shoes more advanced scripting examples where several functions are made.

5.8.1 Catching IO events and exposing them in GATT

This example shows hot to catch I/O events and exposing them via a custom service in GATT data base.

he example service look like the one below and the I/O characteristic has read and notify properties

In order to catch the I/O events and write them to GATT database the following event handled is used in BGScript code.

```
#HW interrupt listener
event hardware_io_port_status(delta, port, irq, state)

# Write I/O status to GATT
call attributes_write(xgatt_io,0,1,irq)
end
```

On DKBLE112 development kit there are buttons in I/O pins P0_0 and P0_1 and in order for this example to work with DKBLE112 the following configuration is needed in hardware.xml.

```
hardware.xml
```

```
<port index="0" pull="down" />
```

5.9 DKBLE112 specific examples

This section contains examples specific to DKBLE112 development kit.

5.9.1 Display initialization

The example below shows how to initialize the display in BLE112 development kit and and how to write data to it.

The supported commands can be found from the displays data sheet as well the initialization sequence.

DKBLE112 display initialization

```
# Boot event listener
event system_boot(major,minor,patch,build,ll_version,protocol,hw)

# Set display to command mode
    call hardware_io_port_write(1,$3,$1)
    call hardware_io_port_config_direction(1,$7)

# Initialize the display (see NHDC0216CZFSWFBW3V3 data sheet)
    call
hardware_spi_transfer(0,11,"\x30\x30\x30\x39\x14\x56\x6d\x70\x0c\x06\x01")

# Set display to data mode
    # Write "Hello world\!" to the display.
    call hardware_io_port_write(1,$3,$3)
    call hardware_spi_transfer(0,12,"Hello world\!")

end
```

1

SPI configuration in *hardware.xml* <usart channel="0" mode="spi_master" alternate="2" polarity="positive" phase="1" endianness="msb" baud="57600" endpoint="none" />

5.9.2 FindMe demo

The example script implements a simple FindMe profile device. The alert status is displayed on BLE112 development kit's display when remote device changes the status.



SPI configuration in hardware.xml <usart channel="0" mode="spi_master" alternate="2" polarity="positive" phase="1" endianness="msb"</pre> baud="57600" endpoint="none" />

DKBLE112 FindMe Target

```
# Boot event listener
event system_boot(major,minor,patch,build,ll_version,protocol,hw)
    # Put display into command mode
    call hardware io port write(1,$3,$1)
    call hardware_io_port_config_direction(1,$7)
    # Configure Display
    call
hardware_spi_transfer(0,11,"\x30\x30\x39\x14\x56\x6d\x70\x0c\x06\x01")
    # Put display into data mode and write
    call hardware_io_port_write(1,$3,$3)
    call hardware_spi_transfer(0,12,"Find Me Demo")
    # Set advertisement parameters according to the Proximity profile. Min
interval 1000ms, max interval 2000ms, use all 3 channels
    call gap_set_adv_parameters(1600, 3200, 7)
    # Start advertisement and enable pairing mode
    call gap_set_mode(gap_general_discoverable,gap_undirected_connectable)
    call sm_set_bondable_mode(1)
end
# Listen for GATT write events
event attributes_value(connection, reason, handle, value_len, value)
    # Put display to command mode and move cursor to position 40
    call hardware io port write(1,$3,$1)
    call hardware_spi_transfer(0,1,"\xc0")
    #display to data mode
    call hardware_io_port_write(1,$3,$3)
    # Read value and enable corresponding alert
    level=value(0:1)
    if level=0 then
       call hardware_spi_transfer(0,10,"No Alert ")
    end if
    if level=1 then
       call hardware spi transfer(0,10,"Mild Alert")
    end if
    if level=2 then
        call hardware_spi_transfer(0,10,"High Alert")
    end if
end
# Disconnection event listener
event connection_disconnected(handle,result)
    # Restart advertisement
    call gap_set_mode(gap_general_discoverable,gap_undirected_connectable)
end
```

5.9.3 Temperature and battery readings to display

The example below shows how to initialize the display in BLE112 development kit and and how to write temperature and battery (using potentiometer) readings into it.

The supported commands can be found from the displays data sheet as well the initialization sequence.

```
A
```

SPI configuration in *hardware.xml*<usart channel="0" mode="spi_master" alternate="2" polarity="positive" phase="1" endianness="msb" baud="57600" endpoint="none" />

DKBLE112 display, battery and temperature sensors

```
dim string(3)
dim milliv
\dim tmp(4)
dim offset
dim celsius
# Boot event listener
event system_boot(major,minor,patch,build,ll_version,protocol,hw)
 # Initialize the display (see NHD-C0216CZ-FSW-FBW-3V3 data sheet)
 call hardware_io_port_write(1,$7,$1)
 call hardware_io_port_config_direction(1,$7)
 call
hardware\_spi\_transfer(0,11,"\x30\x30\x39\x14\x56\x6d\x70\x0c\x06\x01")
 call hardware_io_port_write(1,$7,$3)
 # Write "Batt.: " to the display.
 call hardware_spi_transfer(0,7,"Batt.: ")
 # Change display data address
 call hardware io port_write(1,$7,$1)
 call hardware_spi_transfer(0,1,"\xc0")
 # Write "Temp.: " to the displays 2nd line
 call hardware_io_port_write(1,$7,$3)
 call hardware_spi_transfer(0,7,"Temp.: ")
 # Start timer @ ~2sec interval
call hardware_set_soft_timer($ffff, 0 ,0)
# Timer event listener
event hardware_soft_timer(handle)
 #read potentiometer for battery
 call hardware_adc_read(6,1,2)
#read internal temperature
call hardware_adc_read(14,3,0)
end
```

DKBLE112 display, battery and temperature sensors (CONTINUED)

```
#ADC event listener
event hardware_adc_result(input,value)
 # Received battery reading
 if (input = 6) then
        #Convert HEX to STRING
    milliv=value/11+8
  tmp(0:1) = (milliv/1000) + (milliv / 10000*-10) + 48
  tmp(1:1) = (milliv/100) + (milliv / 1000*-10) + 48
  tmp(2:1) = (milliv/10) + (milliv / 100*-10) + 48
  tmp(3:1) = (milliv) + (milliv / 10*-10) + 48
  # Change display data address
  call hardware_io_port_write(1,$7,$1)
  call hardware_spi_transfer(0,1,"\x87")
  # Write battery value
  call hardware_io_port_write(1,$7,$3)
  call hardware_spi_transfer(0,4,tmp(0:4))
  call hardware_spi_transfer(0,3," mV")
end if
 # Received temperature reading
 if (input = 14) then
 offset=-1490
  # ADC value is 12 MSB
  celsius = value / 16
  # Calculate temperature
  # ADC*V_ref/ADC_max / T_coeff + offset
  celsius = (10*celsius*1150/2047) * 10/45 + offset
  #Convert HEX to STRING
  string(0:1) = (celsius / 100) + 48
  string(1:1) = (celsius / 10) + (celsius / -100 * 10) + 48
  string(2:1) = celsius + (celsius / 10 * -10) + 48
  # Change display data address
  call hardware_io_port_write(1,$7,$1)
  call hardware_spi_transfer(0,1,"\xc7")
  # Write temperature value
  call hardware_io_port_write(1,$7,$3)
  call hardware spi transfer(0,2,string(0:2))
  call hardware_spi_transfer(0,1,".")
  call hardware_spi_transfer(0,1,string(2:1))
  call hardware_spi_transfer(0,1,"\xf2")
  call hardware_spi_transfer(0,1,"C")
end if
end
```

5.10 BGScript tricks

5.10.1 HEX to ASCII

Printing local BT address on the display in DKBLE112

```
dim t(12)
dim addr(6)
event system_boot(major,minor,patch,build,ll_version,protocol,hw)
 call hardware_io_port_write(1,$7,$1)
 call hardware_io_port_config_direction(1,$7)
 #Initialize the display
 call
hardware_spi_transfer(0,11,"\x30\x30\x30\x39\x14\x56\x6d\x70\x0c\x06\x01")
 call hardware_io_port_write(1,$7,$3)
 #Get local BT address
 call system_address_get( )(addr(0:6))
 t(0:1) = (addr(5:1)/\$10) + 48 + ((addr(5:1)/\$10)/10*7)
 t(1:1) = (addr(5:1)&$f) + 48 + ((addr(5:1)&$f)/10*7)
 t(2:1) = (addr(4:1)/\$10) + 48 + ((addr(4:1)/\$10)/10*7)
 t(3:1) = (addr(4:1)&f) + 48 + ((addr(4:1)&f)/10*7)
 t(4:1) = (addr(3:1)/\$10) + 48 + ((addr(3:1)/\$10)/10*7)
 t(5:1) = (addr(3:1)&$f) + 48 + ((addr(3:1)&$f)/10*7)
 t(6:1) = (addr(2:1)/\$10) + 48 + ((addr(2:1)/\$10)/10*7)
 t(7:1) = (addr(2:1)&$f) + 48 + ((addr(2:1)&$f)/10*7)
 t(8:1) = (addr(1:1)/\$10) + 48 + ((addr(1:1)/\$10)/10*7)
 t(9:1) = (addr(1:1)&f) + 48 + ((addr(1:1)&f)/10*7)
 t(10:1) = (addr(0:1)/\$10) + 48 + ((addr(0:1)/\$10)/10*7)
 t(11:1) = (addr(0:1)&$f) + 48 + ((addr(0:1)&$f)/10*7)
 call hardware_spi_transfer(0,12,t(0:12))
end
```

5.10.2 UINT to ASCII

To display sensor readings in the display, integer values must be converted to ASCII. Currently there is no build-in function for doing this in the BGScript, but the following function can be used to convert integers to ASCII:

```
a = (rh / 100)
b = (rh / 10) + (rh / -100 * 10)
c = rh + (rh / 10 * -10)
And as BGScript code:
```

Converting 3 digit interger to ASCII

```
dim data
dim string(3)

string(0:1) = (data / 100) + 48
string(1:1) = (data / 10) + (data / -100 * 10) + 48
string(2:1) = data + (data / 10 * -10) + 48
```

To present the string in the display of the evaluation kit please refer to DKBLE112 display initialization -- BGScript

6 BGScript editors

This section contains different tips and tricks for editors and IDEs.

6.1 Notepad ++

Notepad++ is very flexible text editor for programming purposes. Application and documentation can be downloaded from http://notepad-plus-plus.org/.

6.1.1 Syntax highlight for BGScript

Notepad++ doesn't currently contain syntax highlighting for BGScript by default. You can however download syntax highlighting rules defined by Bluegiga.

Installing the BGScript syntax highlight rules into Notepad++ is easy:

- 1. Download the syntax highlighting rules from http://techforum.bluegiga.com/ble112/
- 2. Import the highlighting rules to Notepad++: View->User-Defined Dialogue->Import.
- 3. When editing the code, enable syntax highlighting from : Language -> BGscript



7 Contact information

Sales: sales@bluegiga.com

Technical support: support@bluegiga.com

http://techforum.bluegiga.com

Orders: orders@bluegiga.com

WWW: http://www.bluegiga.com

http://www.bluegiga.hk

Head Office / Finland: Phone: +358-9-4355 060

Fax: +358-9-4355 0660

Sinikalliontie 5 A 02630 ESPOO

FINLAND

Head address / Finland: P.O. Box 120

02631 ESPOO

FINLAND

Sales Office / USA: Phone: +1 770 291 2181

Fax: +1 770 291 2183

Bluegiga Technologies, Inc.

3235 Satellite Boulevard, Building 400, Suite 300

Duluth, GA, 30096, USA

Sales Office / Hong-Kong: Phone: +852 3182 7321

Fax: +852 3972 5777

Bluegiga Technologies, Inc.

Unit 10-18, 32/F, Tower 1, Millennium City 1, 388 Kwun Tong Road, Kwun Tong, Kowloon,

Hong Kong